
BASIC LIGHTNING 

User Manual 
by David Hunter 

INTRODUCTION 

BASIC LIGHTNING is an extension to the CBM64's resident BASIC interpreter which 
adds over 150 ccmmands. The additional ccmmands cover three main areas: 
structured programming, graphics and sound. One of BASIC LIGHTNING'S most 
powerful features is the ability to multi-task; up to five parts of a BASIC 
program can be run at once. A compiler for BASIC LIGHTNING will be available in 
early 1985 which will produce "stand-alone" compiled programs which can be 
marketed without paying royalties. 

The structured programming ccmmands included in BASIC LIGHTNING include all the 
control ccmmands found in PASCAL. This includes multiple-line IF-THEN-ELSE, 
REPEAT-UNTIL, WHILE-WEND as well as PROCEDURES with full parameter passing 
(including arrays), multiple-line functions and CASE-OF. 

The sound and graphics ccmmands included are the same as those in WHITE LIGHTNING, 
adapted to BASIC syntax, and include ccmmands to plot points, draw lines, move 
software sprites to and frcm the screen, scroll blocks of the screen or sprites 
with character or pixel resolution, rotate, invert or mirror sprites or screen 
windows and exchange sprites with parts of the screen. You can also make use of 
the 64's inbuilt hardware sprites and it is possible to redefine the 64's 
character set. 

To use this manual, it is essential that you already understand how to program 
using the 64's resident BASIC interpreter. 

LOADING BASIC LIGHTNING 

To load BASIC LIGHTNING frcm tape, type shift-run/stop and start the tape in the 
normal way. To load it frcm disk, type LOAD "BL",8,1. 

1. STRUCTURED PROGRAMMING 

The structured programming ccmmands provided will allow you to program without 
using the "GOTO" statement and should result in programs which are easier to 
understand and modify. 

Before proceeding, please note that commas are used in the LIST command instead of 
minus signs: for example, use LIST 100,200 instead of LIST 100-200. Also, 
keywords are no longer abbreviated using the shift key. Instead, a full stop is 
used. For example, "LIST" is abbreviated to "LI." not "L shift I". A list of 
all the keywords and abbreviations can be found in Appendix B. 

Also, you can temporarily halt a listing by using the spacebar - press it again to 
re-start. 

IF-THEN-ELSE 

This is a simple extension to the existing IF-THEN statement. If the condition is 
true, the statements between THEN and ELSE are executed, otherwise the statements 
from the ELSE until the end of the line are executed. The ELSE can, of course, be 
emitted. 

Putting more than one IF-THEN-ELSE on a line is allowed, for example: 

IF a THEN PRINT 1 ELSE IF b THEN PRINT 2 ELSE PRINT 3 

1 is printed if 'a' is true, irrespective of the value of 'b'. 
2 is printed if 'a' is false and 'b' is true. 
3 is printed if 'a' is false and 'b1 is false. 

However, the following is ambiguous: 

IF a THEN IF b PRINT 1 ELSE PRINT 2 

1 is printed if 'a' is true and 'b' is true. 
2 is printed if 'a' is true and 'b' is false. 

- the ELSE statement is assumed to belong to the most recent IF. 

The THEN can be omitted if it is directly followed by a command (rather than an 
assignment), e.g. 

IF 1=6 PRINT "UNDEFINED" 

CIF-CELSE-CEND 

This is a more general version of IF-THEN-ELSE: it can be spread over any number 
of lines. The CIF is equivalent to IF, CELSE is equivalent to ELSE and CEND is 
used to mark the end of the statement. CIF, CELSE and CEND are separated from 
other statements by colons if more than one statement is put on the line. 

Example: 

100 LABEL stcheck 'subroutine to check ST 
110 ' 
120 CIF stOO 
130 OPEN 15,8,15 
140 INPUT#15,w$,x$,y$,z$ 
150 PRINT w$;",";x$;",";y$;",";z$ 
160 er=l 
170 CLOSE 15 
180 CELSE 
190 er=0 
200 CEND 
210 RETURN 

If you have typed in any of the examples so far, you will have noticed that all 
reserved words are printed in upper case when listing, and everything else is 
printed in lower case. This makes the program easier to read, particularly if 
many statements are put on one line without any spaces. 

In the above listing, you will also see that indentation of two spaces has been 
used for statements inside the CIF-CEND. This makes it possible to see at a 
glance what the structure of a program is, although it uses up a lot of memory. 
Also, single quotes have been used for on-line carmenting. A single quote is 
equivalent to ":REM". There is a LABEL statement in line 100 - this is dealt with 
next. 

LABELS 

The use of labels enables subroutines and data to be given symbolic names rather 
than being referenced by meaningless line numbers. 

A label is defined by using the LABEL statement - for an example, look at the 
previous listing. The subroutine in that example would be called using "GOSUB 
stcheck". 

The label itself must start with a letter, the rest of it consisting of letters, 
numbers, "$" and "%" signs. For example, the following are all legal labels: 

z9 grid3% i3vl v$62 

Unlike ordinary variables, all the characters are significant, not just the first 
two. Also, there is no restriction on the use of reserved words in labels; it 
would be perfectly legal to define a label called 'print' or 'teletype' for 
example. There is one exception to this rule; 'ELSE' cannot be included. 

The RESTORE command has been extended so that it can be used with a line number or 
label. Also, ON-RESTORE can be used in the same way as ON-GOTO or ON-QOSUB. 

Example: 

10 INPUT "Skill level (1,2 or 3)";i 
20 ON i RESTORE bl,wl,ml 
30 READ a$ 
40 PRINT "loading ";a$;"..." 
50 READ b$ 
60 LOAD b$,8,l 
70 LABEL bl:DATA "basic lightning","bl" 
80 LABEL wl:DATA "white lightning","wl" 
90 LABEL ml:DATA "machine lightning","ml" 

Labels are also used to define procedures and multiple-line functions; this is 
explained fully later. 

REPEAT-UNTIL 

REPEAT and UNTIL are used to set up a loop where it is required to repeat a set of 
statements at least once. The REPEAT command is put at the top of the loop, and 
the UNTIL is at the bottom of the loop. Every time the UNTIL is encountered, the 
expression after it is evaluated, and if it gives a FALSE value, the computer goes 
back to the corresponding REPEAT and starts the loop again. 

To see how this works, type in the following and run it: 



10 REPEAT 
20 INPUT "4-charac te r code";a$ 
30 UNTIL LEN(a$)=4 

The computer will keep prompting for the string a$ until you type in a string of 
the correct length - four characters. 

WHILE-WEND 

WHILE and WEND are used to set up a loop where it is required to repeat a set of 
statements zero or more times. This is quite similar to the REPEAT-UNTIL loop; 
however, the test for exiting the loop is made at the top, not the bottom. 
Therefore, if the condition is not met when first entering the loop, it will not 
be executed at all. 

Try running this program: 

10 INPUT x$ 
20 WHILE x$<>"" 
30 PRINT x$ 
40 x$=RIGHT$(x$,LEN(x$)-l) 
50 WEND 
60 PRINT 
70 PRINT" * * * end * * * " 

If you type in "while-wend", the following will be printed: 

while-wend 
hile-wend 
ile-wend 
le-wend 
e-wend 
-wend 
wend 
end 
nd 
d 

"k ic ~k £nrj k it ic 

However, if you type in a null string (two quotes and RETURN), you will see that 
the loop itself is not executed at all since x$<>"" gives a FALSE value the first 
time around. 

Since TRUE and FALSE return values of -1 and 0 respectively, it is possible to set 
up an infinite loop using: 

REPEAT UNTIL FALSE 
or WHILE TRUE WEND 

EXIT 

Sometimes it is convenient to exit from a loop prematurely, and the EXIT command 
is provided to do this. For example, let's look at the example that was used 
before for the REPEAT-UNTIL loop: 



10 REPEAT 
20 INPUT "4-character code";a$ 
30 UNTIL LEN(a$)=4 

Suppose that we want to give an error message if the code is not 4 characters 
long: 

10 REPEAT 
20 INPUT "4-character code";a$ 
25 IF LEN(a$)<>4 PRINT "4 characters please - try again" 
30 UNTIL LEN(a$)=4 

This could be more efficiently coded using EXIT, since in the above, the test for 
the correct length of a$ has to be made twice: 

10 REPEAT 
20 INPUT "4-character ccde";a$ 
25 IF LEN(a$)=4 EXIT 
27 PRINT "4 characters please - try again" 
30 UNTIL FALSE 

If the loop conditions are more complex, EXIT can save a lot of re-typing. EXIT 
can also be used to leave WHILE-WEND and FOR-NEXT loops. 

NOTE: It is not advisable to jump out of a loop using a GOTO statement as this 
corrupts the stack and may cause error messages later in the program. 

CASE-OF-CASEND 

The CASE statement is used to select between a number of alternative courses of 
action, depending on the value of a variable which may be either numerical or 
string. 

The CASE is used at the top of the statement and is followed by the variable to be 
tested. After each OF there is a list of values, separated by commas. If the 
variable being tested is equal to one of these expressions, the code up until the 
next OF or CASEND is executed, otherwise it is ignored. The CASEND is put at the 
end of the CASE statement. "OF OR" is interpreted as an instruction to execute 
the following code if none of the statements so far have been executed. 

Example: 

10 INPUT a 
20 CASE a 
30 OF 3:PRINT "Three French hens," 
40 OF 2,3:PRINT "Two turtle doves," 
50 OF 1,2,3:PRINT "And a partridge in a pear tree." 
60 OF OR : STOP 
70 CASEND 

PROCEDURES 

Procedures, like subroutines, are sections of code which are used several times in 
a program. However, they are much more powerful than subroutines because it is 
possible to pass parameters to them. 

Clear the memory using NEW, then type in the following: 

10 LABEL test(i) 
20 LOCAL j 
30 FDR j=l to i 
40 PRINTj; 
50 NEXTj 
60 PRINT 
70 PROCEND 

(the computer will automatically convert the keywords to upper case). 

Now type " proctest(lO) ", and 1 2 3 4 5 6 7 8 9 10 is printed. 

The "10" in proctest(lO) is called an actual parameter, and the variable i is 
called the formal parameter. When the procedure is called, the formal parameter 
is made equal to the actual parameter, and the body of the procedure is executed. 
All formal parameters are local to the procedure - this means that they are 
separate from any variables of the same name that are used in the rest of the 
program. It is also possible to create additional local variables for use within 
a procedure using the LOCAL command, as in line 20. If there is a variable called 
j in the main program, it will not be altered by the procedure because j is local. 
To prove this, type the following: 

for j=l to 12 : proctest(j) : next 

1 
1 2 
1 2 3 
1 2 3 4 
1 2 3 4 5 
1 2 3 4 5 6 
1 2 3 4 5 6 7 
1 2 3 4 5 6 7 8 
1 2 3 4 5 6 7 8 9 
1 2 3 4 5 6 7 8 9 10 
1 2 3 4 5 6 7 8 9 10 11 
1 2 3 4 5 6 7 8 9 10 11 12 

As you can see, the two variable j's are separate and do not interfere with one 
another. 

Parameter passing is not restricted to reals - integers and strings can be passed 
as well. Any expression can be used as an actual parameter. 

To illustrate passing strings as parameters, type in the following procedure: 

100 LABEL centre (a$,i) 
110 PRINT@0,i;" " '40 spaces 
120 PRINT@20 - LEN(a$)/2,i;a$ 
130 PROCEND 

Note that this procedure has two parameters instead of one as in the previous 
example - the number of parameters a procedure can have is only limited by the 
maximum line length of 80 characters. 

In lines 110 and 120, PRINTS is used: PRINT@X,Y; will position the cursor on 
screen column X and row Y before printing the data following it. 

Type 'proccentre ("basic lightning",10)', and "basic lightning" will be printed on 
the middle of the tenth line of the screen. 

The LOCAL command can be used to create more than one variable by separating the 
names by commas. Local integers and strings can be defined also, and if more 
variables are required than can fit on one line, several LOCAL commands can be 
used on separate consecutive lines. LOCAL, if used, is normally the first command 
in the procedure. 

VAR Parameters 

In the examples you have seen so far, none of the procedures have had to alter any 
of the actual parameters. Indeed, you will find that if you modify any of the 
previous examples so that a formal parameter is changed inside the procedure, the 
corresponding actual parameter will remain unaltered. 

If a procedure is to modify any of its parameters, they must be declared as VAR 
parameters in the procedure heading, like this: 

100 LABEL inp(A$,VAR B$) 
110 PRINT@0,10;a$; 
120 LNPUTb$ 
130 PRINT@0,10;" " '40 spaces 
140 PROCEND 

Now clear the screen using SHIFT-CLR and type: 

PROCinp ("Name",x$) 

The computer will execute the procedure "inp" and input a string from the 
keyboard. After this, if you type ?x$ you will see that x$ is equal to what you 
typed in, i.e. the formal parameter b$ was altered inside the procedure and this 
has resulted in x$ being changed as well. 

Note that if a parameter is declared in a procedure as a VAR parameter, the 
corresponding actual parameter must be a variable name, not an expression. Also, 
you are not allowed to pass single elements of arrays as VAR parameters. 

Arrays as Parameters 

You can also pass whole arrays as parameters to procedures, although they must be 
declared as VAR parameters. Also, an actual parameter array must have been 
DIMensioned before calling the procedure, even if its dimensions are less than the 
default value of 10. 

When arrays are passed as parameters to procedures, we often do not know what the 
size of the array will be when writing the procedure. For this reason, the SIZE 
function is provided in BASIC LIGHTNING. For example, type in the following: 

DIM x(10,20) 

Now, SIZE(x,0) will give a value of 2, the number of dimensions. SIZE(x,l) gives 
21, the second dimension + 1 and SIZE(x,2) gives 11, the first dimension + 1. 

Here is a simple program which illustrates the use of arrays as parameters: 

10 INPUT n 
20 DIM aS(n) 
30 PROC enter(a$()) 
40 PROC sort(a$()) 
50 PROC out(a$()) 
60 END 
70 ' 
80 LABEL enter (VAR x$()) 
90 LOCAL i 
100 FOR i=l TO SIZE (x$,l)-l 
110 INPUT x$(i) 
120 NEXT i 
130 PROCEND 
140 ' 
150 LABEL sort (VAR y$()) 
160 LOCAL i,j,k$ 
170 REPEAT 
180 j=TRUE 
190 FOR i=l TO SIZE (y$,l)-2 
200 CIF y$ (i) > y$(i+l) 
210 k$=y$(i) 
220 y$(i)=y$(i+l) 
230 y$(i+l)=k$ 
240 j=FALSE 
250 CEND 
260 NEXT i 
270 UNTIL j 
280 PROCEND 
290 ' 
300 LABEL out (VAR z$()) 
310 LOCAL i 
320 FOR i=l TO SIZE (z$,l)-l 
330 PRINT z$(i), 
340 NEXT i 
350 PROCEND 

'get number of records 

'get records 
'sort 
'and print 

'swap two elements 
'if out of sequence 

'until elements in order 

MULTIPLE-LINE FUNCTIONS 

Multiple-line functions are similar to procedures, the difference being that they 
return a value; the end of the function is an "=" sign followed by the value to be 
returned. To call the function, the keyword "CFN" is followed by the label name 
and the parameters inside brackets, as with procedures. As an example, here is a 
function to calculate the factorial of a number: 

10 LABEL fact(n) 
20 LOCAL i,j 
30 i=l 
40 CIF n>l 
50 FOR j=2 TO n 
60 i=i*j 
70 NEXT j 
80 CEND 
90 = i 

When this has been typed in, CFNfact(i) will return the factorial of a number. 
For example, ?CFNfact(4) will print '24'. 

2. MISCELLANEOUS KEYWORDS 

Hexadecimal Numbers 

Hexadecimal numbers can be used in express ions by preceding them with a ' $ ' s ign . 
For example: 

? $9800/2 

will give a result of 19456. 

It is also possible to convert a numerical result into hex using the string 
function HEX$, e.g. ? HEX$ (19456) will give $4C00. The string is always five 
characters long - the '$' plus four digits. The number being converted must be in 
the range 0 to 65535. 

DEEK and DOKE 

These are similar to PEEK and POKE, but they operate on two bytes at once instead 
of one only, using the usual 6502 low byte/high byte order. 

DEEK returns a value between 0 and 65535; for example ? DEEK (65532) will give the 
same result as ? PEEK (65532) + 256 * PEEK (65533). 

DOKE $C000,$55FF is equivalent to POKE $C000,$FF : POKE $C001,$55. 

DISABLE can be used to disable the STOP key. This will only work if it is 
included in a program since it is automatically re-enabled every time you type in 
a line in immediate (or "cattnand") mode. There is no way to break out of a 
program once this command has been used, unless an error occurs. 

DLOAD and DSAVE 

These are very similar to IOAD and SAVE, but they assume that you are using a disk 
drive (device No. 8) rather than a tape recorder (device No. 1). 

Example: DIDAD "SPIGEN" 

PULL 

This command removes the return line number put on the stack by GOSUB from the 
stack. It is useful in menu-driven programs where it is required to move back to 
a higher level menu, in which case the subroutine to do this would be 
PULL:RETURN. 

PI 

PI is a reserved word which will return the value 3.14159265. 

DIR 

If you have a disk drive, DIR can be used to print out the disk directory. As 
with LOAD or SAVE, the filename and device number can be specified; DIR"$0:O*",9 
would print out all filenames on device number 9 which start with "0". For 
further details, refer to your disk drive manual. 

OLD 

This catmand will attempt to restore a program which has been NEWed. If you have 
typed in program lines or created new variables since typing NEW, it will not be 
successful. 

3. GRAPHICS COMMANDS 

The graphics commands included in BASIC LIGHTNING are designed to allow easy 
manipulation of images on and off the screen. This is achieved by carrying out 
operations on a table of sprites which can initially hold up to 8k of data. In 
this context, "sprite" means a graphics character of user-definable dimensions (up 
to 255 characters height or width) which may be displayed on the screen in one of 
the high-resolution modes (either bit-map mode or multi-colour bit-map mcde). 
This is not to be confused with the 64' s own hardware sprites - these can also be 
used from BASIC LIGHTNING and will be dealt with later. 

Each sprite in the table is given a number frcm 1 to 255, and the screen is 
treated as sprite number zero - thus, the same commands can be used for both the 
screen and the sprites. 

SPRITE VARIABLES 

The sprite graphics commands in BASIC LIGHTNING use thirteen variables to pass 
parameters. The variables are: 

SPN 
COL 
ROW 
WID 
HGT 
SPN2 
COL 2 
R0W2 
NUM 
INC 
ATR 
CCOL 
CROW 

sprite no. 1 
column in sprite no. 1 
row in sprite no. 1 
width of window 
height of window 
sprite no. 2 
column in sprite no. 2 
row in sprite no. 2 
number of sides on polygon or number of pixels to scroll 
inclination of polygon 
current attribute 
column for collision detection 
row for collision detection 

(The way that these variables are used will become clear later). 

They can be treated like normal variables; i.e. they can be assigned a value: 

SPN = 3 (Don't use let) 

or used in expressions: 

PRINT SPN-1 

Most of the graphics commands can be used either with or without the parameters 
after them. For example, the command WCLR uses SPN, COL, ROW, WID and HGT as 
parameters. Thus WCLR 0,8,8,4,4 is equivalent to : 

SPN=0:COL=8:ROW=8:WID=4:HGT=4:WCLR 

